Totally Nonnegative Inverse Eigenvalue Problem by Extended discrete Hungry Toda equation

Definition of Algorithm

```
Composition::usage = "This is a value of Output";
Decomposition::usage = "This is a value of Output";
QETable::usage = "This is a value of Output";
TNIEPdhToda::usage = "The function IEPdhTodaTN[]
    returns the square maxtrix with specified eigenvalues.";
TNIEPdhToda::M = "The argument `1` should be a positive integer.";
TNIEPdhToda::NN = "The argument `1` should be a positive integer.";
TNIEPdhToda::lambda = "The argument `1` is NOT List of numeric entries.";
TNIEPdhToda::output =
  "The option Output->`1` should be Composite or Decomposite.";

TNIEPdhToda[NN_Integer, M_Integer, lambda_List,
  param_, OptionsPattern[{Prec → 30, Output → Composition}]] :=
   Module[{m = Length[lambda], f, c, sigma, i, j,
  k, n, e, q, l, L, r, R, A, LR, QE, prec = 30},
      (* check arguments *)
      If[M < 1, Message[IEPdhTodaTN::M, M]];
      If[NN < 1, Message[IEPdhTodaTN::NN, NN]];
      If[! (VectorQ[lambda] && AllTrue[lambda, NumericQ]),
  Message[dTodaIP::lambda, lambda]];
      If[NumericQ[OptionValue[Prec]], prec = OptionValue[Prec]];
      (* eigenvalues *)
      sigma[i_] := sigma[i] = (lambda[[i]]) ^ (1 / (M * NN));
      (* mements *)
      c[i_] := (c[i] = Which[MatchQ[param, _List],
     param[[i]], NumericQ[param], param, True, param[]]);
      f[n_] := f[n] = N[Sum[c[i] * sigma[i] ^ n, {i, 1, m}], prec];
      (* qd-table *)
      e[0, n_] := e[0, n] = 0;
      q[1, n_] := q[1, n] = f[n + NN] / f[n];
      e[k_, n_] := e[k, n] = e[k - 1, n + NN] + q[k, n + M] - q[k, n];
```

```
        q[k_, n_] := q[k, n] = e[k - 1, n + NN] / e[k - 1, n] * q[k - 1, n + M];
        QE = Join[Table[Table[e[k, n], {k, 1, m - 1}], {n, 0, M * (NN - 1), M}],
    Table[Table[q[k, n], {k, 1, m}], {n, NN * (M - 1), 0, -NN}]];
        (* check size *)
        Do[If[e[k, 0] == 0, m = k; Break[]], {k, 1, m}];
        (* generate matrix *)
        r[i_, j_, n_] := q[i, n] /; i == j;
        r[i_, j_, n_] := 1 /; i + 1 == j;
        r[i_, j_, n_] := 0;
        R[n_] := R[n] = Table[r[i, j, n], {i, 1, m}, {j, 1, m}];
        l[i_, j_, n_] := 1 /; i == j;
        l[i_, j_, n_] := e[j, n] /; i - 1 == j;
        l[i_, j_, n_] := 0 /; i != j;
        L[n_] := L[n] = Table[l[i, j, n], {i, 1, m}, {j, 1, m}];
        A = IdentityMatrix[m];
        LR = {};
        Do[A = A.L[n]; AppendTo[LR, L[n]], {n, 0, M * (NN - 1), M}];
        Do[A = A.R[n];
   AppendTo[LR, R[n]], {n, NN * (M - 1), 0, -NN}];
        (* return value *)
        Which[
            OptionValue[Output] === Composition, Return[A],
            OptionValue[Output] === Decomposition, Return[LR],
            OptionValue[Output] === QETable, Return[QE],
            True, Message[IEPdhTodaTN::output, OptionValue[Output]];
   Return[]
        ];
    ]

LowerBidiagonalMatrix[d_, ld_] :=
    Module[{m = 0, i, j, A},
        If[MatchQ[d, _List], m = Max[Length[d], m]];
        If[MatchQ[ld, _List], m = Max[Length[ld] + 1, m]];
        A = Table[Table[Which[
                i == j, If[MatchQ[d, _List], If[i <= Length[d], d[[i]], 0], d],
                i == j + 1,
       If[MatchQ[ld, _List], If[j <= Length[ld], ld[[j]], 0], ld],
                True, 0], {j, 1, m}], {i, 1, m}];
        Return[A];
    ];

UpperBidiagonalMatrix[d_, ud_] :=
    Module[{m = 0, i, j, A},
        If[MatchQ[d, _List], m = Max[Length[d], m]];
        If[MatchQ[ud, _List], m = Max[Length[ud] + 1, m]];
        A = Table[Table[Which[
                i == j, If[MatchQ[d, _List], If[i <= Length[d], d[[i]], 0], d],
```

```
                i + 1 == j,
            If[MatchQ[ud, _List], If[i <= Length[ud], ud[[i]], 0], ud],
                True, 0], {j, 1, m}], {i, 1, m}];
        Return[A];
    ];

IEPdhTodaLRMatrices[NN_Integer, M_Integer, QE_List] :=
    Module[{n, LR},
        LR = Join[Table[LowerBidiagonalMatrix[1, QE[[n]]], {n, 1, NN}],
            Table[UpperBidiagonalMatrix[QE[[n + NN]], 1], {n, 1, M}]];
        Return[LR];
    ];

IEPdhTodaLRComposition[NN_Integer, M_Integer, LR_List] :=
    Module[{m = 0, n, A},
        Do[m = Max[Length[LR[[n]]], m], {n, 1, NN + M}];
        A = IdentityMatrix[m];
        Do[A = A.LR[[n]], {n, 1, NN}];
        Do[A = A.LR[[NN + n]], {n, 1, M}];
        Return[A];
    ];
```

Sample

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, $c_1 = c_2 = c_3 = c_4 = c_5$=1
Output: totally nonnegative matrix 5-by-5 matrix $A^{(0)}$

```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = 1;
A = TNIEPdhToda[NN, M, lambda, c];
Print["A^(0)=", MatrixForm[N[A, 3]]]
Print["The eigenvalues of generated A^(0) is"];
Eigenvalues[A]
```

$$A^{(0)} = \begin{pmatrix} 3.00 & 3.21 & 1.00 & 0 & 0 \\ 0.612 & 2.69 & 3.17 & 1.00 & 0 \\ 0.0346 & 0.432 & 2.88 & 3.35 & 1.00 \\ 0.000412 & 0.0156 & 0.290 & 3.10 & 3.54 \\ 0 & 0.0000870 & 0.00478 & 0.148 & 3.34 \end{pmatrix}$$

The eigenvalues of generated $A^{(0)}$ is

```
{5.0000000000000000000, 3.9999999999999999999,
 3.0000000000000000000, 1.9999999999999999992, 1.0000000000000000001}
```

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, $c_1 = c_2 = c_3 = c_4 = c_5$=1
Output: lower and upper bidiagonal matrices, $L^{(0)}$, $L^{(M)}$, .., $L^{(M(N-1))}$,

$R^{(N(M-1))}$, .., $R^{(N)}$, $R^{(0)}$

```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = 1;
LR = TNIEPdhToda[NN, M, lambda, c, Output → Decomposition];
Print["L^(0),L^(M),..,L^(M(N-1)),R^(N(M-1)), .., R^(N), R^(0)=", MatrixForm /@ N[LR, 3]]
A = IEPdhTodaLRComposition[NN, M, LR];
Print["A^(0)=", MatrixForm[N[A, 3]]]
Print["The eigenvalues of generated A^(0) is"];
Eigenvalues[A]
```

$L^{(0)}, L^{(M)}, \ldots, L^{(M(N-1))}, R^{(N(M-1))}, \ldots, R^{(N)}, R^{(0)} = \left\{ \begin{pmatrix} 1.00 & 0 & 0 & 0 & 0 \\ 0.0777 & 1.00 & 0 & 0 & 0 \\ 0 & 0.0615 & 1.00 & 0 & 0 \\ 0 & 0 & 0.0363 & 1.00 & 0 \\ 0 & 0 & 0 & 0.0163 & 1.00 \end{pmatrix}, \right.$

$\begin{pmatrix} 1.00 & 0 & 0 & 0 & 0 \\ 0.0681 & 1.00 & 0 & 0 & 0 \\ 0 & 0.0650 & 1.00 & 0 & 0 \\ 0 & 0 & 0.0389 & 1.00 & 0 \\ 0 & 0 & 0 & 0.0172 & 1.00 \end{pmatrix}, \begin{pmatrix} 1.00 & 0 & 0 & 0 & 0 \\ 0.0580 & 1.00 & 0 & 0 & 0 \\ 0 & 0.0675 & 1.00 & 0 & 0 \\ 0 & 0 & 0.0418 & 1.00 & 0 \\ 0 & 0 & 0 & 0.0183 & 1.00 \end{pmatrix},$

$\left. \begin{pmatrix} 1.79 & 1.00 & 0 & 0 & 0 \\ 0 & 1.43 & 1.00 & 0 & 0 \\ 0 & 0 & 1.49 & 1.00 & 0 \\ 0 & 0 & 0 & 1.63 & 1.00 \\ 0 & 0 & 0 & 0 & 1.76 \end{pmatrix}, \begin{pmatrix} 1.68 & 1.00 & 0 & 0 & 0 \\ 0 & 1.42 & 1.00 & 0 & 0 \\ 0 & 0 & 1.54 & 1.00 & 0 \\ 0 & 0 & 0 & 1.67 & 1.00 \\ 0 & 0 & 0 & 0 & 1.79 \end{pmatrix} \right\}$

$A^{(0)} = \begin{pmatrix} 3.00 & 3.21 & 1.00 & 0 & 0 \\ 0.612 & 2.69 & 3.17 & 1.00 & 0 \\ 0.0346 & 0.432 & 2.88 & 3.35 & 1.00 \\ 0.000412 & 0.0156 & 0.290 & 3.10 & 3.54 \\ 0 & 0.0000870 & 0.00478 & 0.148 & 3.34 \end{pmatrix}$

The eigenvalues of generated $A^{(0)}$ is

{5.0000000000000000000, 3.9999999999999999999,
3.0000000000000000000, 1.9999999999999999992, 1.0000000000000000001}

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, $c_1 = c_2 = c_3 = c_4 = c_5 = 1$
Output: variables $q_k^{(n)}$ for $k = 1, 2, \ldots, m$, $n = 0, N, \ldots, N(M-1)$ and $e_k^{(n)}$ for $k = 1, 2, \ldots, m-1$, $n = 0, M, \ldots, M(N-1)$

```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = 1;
QE = TNIEPdhToda[NN, M, lambda, c, Output → QETable];
Print["The table of ek(n) and qk(n) =", MatrixForm[N[QE, 3]]];
LR = IEPdhTodaLRMatrices[NN, M, QE];
A = IEPdhTodaLRComposition[NN, M, LR];
Print["A(0)=", MatrixForm[N[A, 3]]]
Print["The eigenvalues of generated A(0) is"];
Eigenvalues[A]
```

The table of $e_k^{(n)}$ and $q_k^{(n)}$ = $\begin{pmatrix} \{0.0777, 0.0615, 0.0363, 0.0163\} \\ \{0.0681, 0.0650, 0.0389, 0.0172\} \\ \{0.0580, 0.0675, 0.0418, 0.0183\} \\ \{1.79, 1.43, 1.49, 1.63, 1.76\} \\ \{1.68, 1.42, 1.54, 1.67, 1.79\} \end{pmatrix}$

$A^{(0)} = \begin{pmatrix} 3.00 & 3.21 & 1.00 & 0 & 0 \\ 0.612 & 2.69 & 3.17 & 1.00 & 0 \\ 0.0346 & 0.432 & 2.88 & 3.35 & 1.00 \\ 0.000412 & 0.0156 & 0.290 & 3.10 & 3.54 \\ 0 & 0.0000870 & 0.00478 & 0.148 & 3.34 \end{pmatrix}$

The eigenvalues of generated $A^{(0)}$ is

{5.0000000000000000000, 3.9999999999999999999, 3.0000000000000000000, 1.9999999999999999992, 1.0000000000000000001}

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, $c_1 = 1$, c_2=2, c_3=3, c_4=4, c_5=5
Output: totally nonnegative matrix 5-by-5 matrix $A^{(0)}$

```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = {1, 2, 3, 4, 5};
A = TNIEPdhToda[NN, M, lambda, c];
Print["A(0)=", MatrixForm[N[A, 3]]]
Print["The eigenvalues of generated A(0) is"];
Eigenvalues[A]
```

$$A^{(0)} = \begin{pmatrix} 3.67 & 3.41 & 1.00 & 0 & 0 \\ 0.449 & 2.61 & 3.08 & 1.00 & 0 \\ 0.0269 & 0.468 & 2.66 & 3.21 & 1.00 \\ 0.000374 & 0.0204 & 0.326 & 2.90 & 3.42 \\ 0 & 0.000135 & 0.00642 & 0.163 & 3.16 \end{pmatrix}$$

The eigenvalues of generated $A^{(0)}$ is

{5.0000000000000000000, 4.0000000000000000000,
 3.0000000000000000000, 2.0000000000000000000, 1.0000000000000000002}

Inputs: N=3, M=5, m=30, λ_k=k for k=1,2,..,m, c_k = 1 for k = 1, 2, .., m
Parameter: prec=100 (precision of computing)
Output: totally nonnegative matrix 30-by-30 matrix $A^{(0)}$

```
NN = 3;
M = 5;
m = 30;
lambda = Table[i, {i, 1, m}];
c = 1;
A = TNIEPdhToda[NN, M, lambda, c, Prec → 100];
Print["A(0)=", MatrixForm[N[A, 3]]]
Print["The eigenvalues of generated A(0) is"];
Eigenvalues[A]
```

$$A^{(0)} = \left(\begin{array}{cccccccccc}
15.5 & 33.3 & 37.3 & 23.3 & 7.58 & 1.00 & 0 & 0 & 0 & \\
2.08 & 11.3 & 27.2 & 33.6 & 22.2 & 7.45 & 1.00 & 0 & 0 & \\
0.146 & 2.03 & 11.4 & 28.1 & 34.8 & 22.8 & 7.55 & 1.00 & 0 & \\
0.00347 & 0.144 & 1.88 & 11.7 & 29.5 & 36.3 & 23.5 & 7.66 & 1.00 & \\
0 & 0.00325 & 0.118 & 1.75 & 12.0 & 30.8 & 37.7 & 24.1 & 7.77 & 1 \\
0 & 0 & 0.00237 & 0.0982 & 1.64 & 12.4 & 32.1 & 39.1 & 24.7 & 7 \\
0 & 0 & 0 & 0.00175 & 0.0825 & 1.55 & 12.7 & 33.3 & 40.3 & 2 \\
0 & 0 & 0 & 0 & 0.00131 & 0.0699 & 1.46 & 13.1 & 34.5 & 4 \\
0 & 0 & 0 & 0 & 0 & 0.000999 & 0.0595 & 1.37 & 13.4 & 3 \\
0 & 0 & 0 & 0 & 0 & 0 & 0.000766 & 0.0508 & 1.29 & 1 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.000591 & 0.0434 & 1 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.000457 & 0. \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 &
\end{array}\right.$$

The eigenvalues of generated $A^{(0)}$ is

{29.9999999999999999996, 28.9999999999999999995, 27.9999999999999999997, 26.9999999999999999996, 25.9999999999999999992, 24.9999999999999999997, 23.9999999999999999993, 22.9999999999999999998, 21.9999999999999999997, 20.9999999999999999999, 19.9999999999999999994, 18.9999999999999999997, 17.9999999999999999997, 16.9999999999999999998, 15.9999999999999999997, 14.9999999999999999992, 13.9999999999999999998, 12.9999999999999999982, 12.0000000000000000011, 10.9999999999999999975, 9.99999999999999999537, 9.00000000000000000025, 7.99999999999999999977, 6.99999999999999999982, 5.99999999999999999997, 5.00000000000000000262, 4.00000000000000000090, 3.00000000000000000012, 1.99999999999999999824, 0.999999999999999998126}