Totally Nonnegative Inverse Eigenvalue Problem by Extended discrete Hungry Toda equation

Definition of Algorithm

In[1]:=
```
Composition::usage = "This is a value of Output";
Decomposition::usage = "This is a value of Output";
QETable::usage = "This is a value of Output";
TNIEPdhToda::usage = "The function IEPdhTodaTN[]
    returns the square maxtrix with specified eigenvalues.";
TNIEPdhToda::M = "The argument `1` should be a positive integer.";
TNIEPdhToda::NN = "The argument `1` should be a positive integer.";
TNIEPdhToda::lambda = "The argument `1` is NOT List of numeric entries.";
TNIEPdhToda::output =
  "The option Output->`1` should be Composite or Decomposite.";

TNIEPdhToda[NN_Integer, M_Integer, lambda_List,
  param_, OptionsPattern[{Output → Composition}]] :=
   Module[{m = Length[lambda], f, c, sigma, i, j, k, n, e, q, l, L, r, R, A, LR, QE},
      (* check arguments *)
      If[M < 1, Message[IEPdhTodaTN::M, M]];
      If[NN < 1, Message[IEPdhTodaTN::NN, NN]];
      If[! (VectorQ[lambda] && AllTrue[lambda, NumericQ]),
   Message[dTodaIP::lambda, lambda]];
      (* eigenvalues *)
      sigma[i_] := sigma[i] = (lambda[[i]]) ^ (1 / (M * NN));
      (* mements *)
      c[i_] := (c[i] = Which[MatchQ[param, _List],
     param[[i]], NumericQ[param], param, True, param[]]);
      f[n_] := f[n] = N[Sum[c[i] * sigma[i] ^ n, {i, 1, m}]];
      (* qd-table *)
      e[0, n_] := e[0, n] = 0;
      q[1, n_] := q[1, n] = f[n + NN] / f[n];
      e[k_, n_] := e[k, n] = e[k - 1, n + NN] + q[k, n + M] - q[k, n];
      q[k_, n_] := q[k, n] = e[k - 1, n + NN] / e[k - 1, n] * q[k - 1, n + M];
      QE = Join[Table[Table[e[k, n], {k, 1, m - 1}], {n, 0, M * (NN - 1), M}],
```

```
   Table[Table[q[k, n], {k, 1, m}], {n, NN * (M - 1), 0, -NN}]];
      (* check size *)
      Do[If[e[k, 0] == 0, m = k; Break[]], {k, 1, m}];
      (* generate matrix *)
      r[i_, j_, n_] := q[i, n] /; i == j;
      r[i_, j_, n_] := 1 /; i + 1 == j;
      r[i_, j_, n_] := 0;
      R[n_] := R[n] = Table[r[i, j, n], {i, 1, m}, {j, 1, m}];
      l[i_, j_, n_] := 1 /; i == j;
      l[i_, j_, n_] := e[j, n] /; i - 1 == j;
      l[i_, j_, n_] := 0 /; i != j;
      L[n_] := L[n] = Table[l[i, j, n], {i, 1, m}, {j, 1, m}];
      A = IdentityMatrix[m];
      LR = {};
      Do[A = A.L[n]; AppendTo[LR, L[n]], {n, 0, M * (NN - 1), M}];
      Do[A = A.R[n];
  AppendTo[LR, R[n]], {n, NN * (M - 1), 0, -NN}];
      (* return value *)
      Which[
          OptionValue[Output] === Composition, Return[A],
          OptionValue[Output] === Decomposition, Return[LR],
          OptionValue[Output] === QETable, Return[QE],
          True, Message[IEPdhTodaTN::output, OptionValue[Output]];
  Return[]
      ];
   ]

LowerBidiagonalMatrix[d_, ld_] :=
   Module[{m = 0, i, j, A},
      If[MatchQ[d, _List], m = Max[Length[d], m]];
      If[MatchQ[ld, _List], m = Max[Length[ld] + 1, m]];
      A = Table[Table[Which[
              i == j, If[MatchQ[d, _List], If[i <= Length[d], d[[i]], 0], d],
              i == j + 1,
      If[MatchQ[ld, _List], If[j <= Length[ld], ld[[j]], 0], ld],
              True, 0], {j, 1, m}], {i, 1, m}];
      Return[A];
   ];

UpperBidiagonalMatrix[d_, ud_] :=
   Module[{m = 0, i, j, A},
      If[MatchQ[d, _List], m = Max[Length[d], m]];
      If[MatchQ[ud, _List], m = Max[Length[ud] + 1, m]];
      A = Table[Table[Which[
              i == j, If[MatchQ[d, _List], If[i <= Length[d], d[[i]], 0], d],
              i + 1 == j,
      If[MatchQ[ud, _List], If[i <= Length[ud], ud[[i]], 0], ud],
```

```
                True, 0], {j, 1, m}], {i, 1, m}];
        Return[A];
    ];

IEPdhTodaLRMatrices[NN_Integer, M_Integer, QE_List] :=
    Module[{n, LR},
        LR = Join[Table[LowerBidiagonalMatrix[1, QE[[n]]], {n, 1, NN}],
            Table[UpperBidiagonalMatrix[QE[[n + NN]], 1], {n, 1, M}]];
        Return[LR];
    ];

IEPdhTodaLRComposition[NN_Integer, M_Integer, LR_List] :=
    Module[{m = 0, n, A},
        Do[m = Max[Length[LR[[n]]], m], {n, 1, NN + M}];
        A = IdentityMatrix[m];
        Do[A = A.LR[[n]], {n, 1, NN}];
        Do[A = A.LR[[NN + n]], {n, 1, M}];
        Return[A];
    ];
```

Sample

Inputs: N=3, M=2, m=5, $\lambda_1=1$, $\lambda_2=2$, $\lambda_3=3$, $\lambda_4=4$, $\lambda_5=5$, $c_1 = c_2= c_3= c_4 = c_5=1$
Output: totally nonnegative matrix 5-by-5 matrix $A^{(0)}$

```
In[14]:= NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = 1;
A = TNIEPdhToda[NN, M, lambda, c];
Print["A^(0)=", MatrixForm[A]]
Print["The eigenvalues of generated A^(0) is"];
Eigenvalues[A]
```

$$A^{(0)} = \begin{pmatrix} 3. & 3.21358 & 1. & 0. & 0. \\ 0.611584 & 2.68722 & 3.16666 & 1. & 0. \\ 0.0346236 & 0.431591 & 2.87648 & 3.35385 & 1. \\ 0.000411598 & 0.01557 & 0.290064 & 3.10084 & 3.53921 \\ 0. & 0.0000869724 & 0.00478285 & 0.147553 & 3.33546 \end{pmatrix}$$

The eigenvalues of generated $A^{(0)}$ is

```
Out[21]= {5., 4., 3., 2., 1.}
```

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, $c_1 = c_2 = c_3 = c_4 = c_5$=1
Output: lower and upper bidiagonal matrices, $L^{(0)}$, $L^{(M)}$, .., $L^{(M(N-1))}$, $R^{(N(M-1))}$, .., $R^{(N)}$, $R^{(0)}$

In[22]:=
```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = 1;
LR = TNIEPdhToda[NN, M, lambda, c, Output → Decomposition];
Print["L^(0),L^(M),..,L^(M(N-1)),R^(N(M-1)), .., R^(N), R^(0)=", MatrixForm /@ LR]
A = IEPdhTodaLRComposition[NN, M, LR];
Print["A^(0)=", MatrixForm[A]]
Print["The eigenvalues of generated A^(0) is"];
Eigenvalues[A]
```

$L^{(0)}, L^{(M)}, .., L^{(M(N-1))}, R^{(N(M-1))},\ ..,\ R^{(N)},\ R^{(0)} =$

$$\left\{\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0.0776857 & 1 & 0 & 0 & 0 \\ 0 & 0.0615495 & 1 & 0 & 0 \\ 0 & 0 & 0.036343 & 1 & 0 \\ 0 & 0 & 0 & 0.0162954 & 1 \end{pmatrix},\right.$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0.0681299 & 1 & 0 & 0 & 0 \\ 0 & 0.0650369 & 1 & 0 & 0 \\ 0 & 0 & 0.0388824 & 1 & 0 \\ 0 & 0 & 0 & 0.0172248 & 1 \end{pmatrix},$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0.0580459 & 1 & 0 & 0 & 0 \\ 0 & 0.067549 & 1 & 0 & 0 \\ 0 & 0 & 0.0417606 & 1 & 0 \\ 0 & 0 & 0 & 0.0182778 & 1 \end{pmatrix},$$

$$\begin{pmatrix} 1.78948 & 1 & 0 & 0 & 0 \\ 0 & 1.42693 & 1 & 0 & 0 \\ 0 & 0 & 1.49085 & 1 & 0 \\ 0 & 0 & 0 & 1.6321 & 1 \\ 0 & 0 & 0 & 0 & 1.7631 \end{pmatrix},$$

$$\left.\begin{pmatrix} 1.67647 & 1 & 0 & 0 & 0 \\ 0 & 1.4241 & 1 & 0 & 0 \\ 0 & 0 & 1.53586 & 1 & 0 \\ 0 & 0 & 0 & 1.66886 & 1 \\ 0 & 0 & 0 & 0 & 1.79012 \end{pmatrix}\right\}$$

$$A^{(0)} = \begin{pmatrix} 3. & 3.21358 & 1. & 0. & 0. \\ 0.611584 & 2.68722 & 3.16666 & 1. & 0. \\ 0.0346236 & 0.431591 & 2.87648 & 3.35385 & 1. \\ 0.000411598 & 0.01557 & 0.290064 & 3.10084 & 3.53921 \\ 0. & 0.0000869724 & 0.00478285 & 0.147553 & 3.33546 \end{pmatrix}$$

The eigenvalues of generated $A^{(0)}$ is

Out[31]= {5., 4., 3., 2., 1.}

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, $c_1 = c_2 = c_3 = c_4 = c_5$=1

Output: variables $q_k^{(n)}$ for $k = 1, 2, .., m,\ n = 0, N, .., N(M-1)$ and $e_k^{(n)}$ for $k = 1, 2, ..., m-1,\ n = 0, M, .., M(N-1)$

In[32]:=
```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = 1;
QE = TNIEPdhToda[NN, M, lambda, c, Output → QETable];
Print["The table of ek(n) and qk(n) =", MatrixForm[QE]];
LR = IEPdhTodaLRMatrices[NN, M, QE];
A = IEPdhTodaLRComposition[NN, M, LR];
Print["A(0)=", MatrixForm[A]]
Print["The eigenvalues of generated A(0) is"];
Eigenvalues[A]
```

The table of $e_k{}^{(n)}$ and $q_k{}^{(n)}$ =
$$\begin{pmatrix} \{0.0776857, 0.0615495, 0.036343, 0.0162954\} \\ \{0.0681299, 0.0650369, 0.0388824, 0.0172248\} \\ \{0.0580459, 0.067549, 0.0417606, 0.0182778\} \\ \{1.78948, 1.42693, 1.49085, 1.6321, 1.7631\} \\ \{1.67647, 1.4241, 1.53586, 1.66886, 1.79012\} \end{pmatrix}$$

$$A^{(0)} = \begin{pmatrix} 3. & 3.21358 & 1. & 0. & 0. \\ 0.611584 & 2.68722 & 3.16666 & 1. & 0. \\ 0.0346236 & 0.431591 & 2.87648 & 3.35385 & 1. \\ 0.000411598 & 0.01557 & 0.290064 & 3.10084 & 3.53921 \\ 0. & 0.0000869724 & 0.00478285 & 0.147553 & 3.33546 \end{pmatrix}$$

The eigenvalues of generated $A^{(0)}$ is

Out[42]= {5., 4., 3., 2., 1.}

Inputs: N=3, M=2, m=5, λ_1=1, λ_2=2, λ_3=3, λ_4=4, λ_5=5, c_1 = 1, c_2=2, c_3=3, c_4 =4, c_5=5

Output: totally nonnegative matrix 5-by-5 matrix $A^{(0)}$

In[43]:=
```
NN = 3;
M = 2;
lambda = {1, 2, 3, 4, 5};
c = {1, 2, 3, 4, 5};
A = TNIEPdhToda[NN, M, lambda, c];
Print["A(0)=", MatrixForm[A]]
Print["The eigenvalues of generated A(0) is"];
Eigenvalues[A]
```

$$A^{(0)} = \begin{pmatrix} 3.66667 & 3.40772 & 1. & 0. & 0. \\ 0.448598 & 2.61473 & 3.07978 & 1. & 0. \\ 0.0268599 & 0.468089 & 2.66143 & 3.21151 & 1. \\ 0.000373697 & 0.0204249 & 0.3256 & 2.90101 & 3.42222 \\ 0. & 0.000135082 & 0.00641783 & 0.163483 & 3.15616 \end{pmatrix}$$

The eigenvalues of generated $A^{(0)}$ is

Out[50]= {5., 4., 3., 2., 1.}